# This file is part of BlackArch Linux ( https://www.blackarch.org/ ).
# See COPYING for license details.

pkgname=mp3nema
pkgver=0.4
pkgrel=3
groups=('blackarch' 'blackarch-forensic')
pkgdesc='A tool aimed at analyzing and capturing data that is hidden between frames in an MP3 file or stream, otherwise noted as "out of band" data.'
arch=('x86_64' 'aarch64')
url='http://packetstormsecurity.com/files/76432/MP3nema-Forensic-Analysis-Tool.html'
license=('GPL3')
depends=('glibc')
source=("https://dl.packetstormsecurity.net/forensics/mp3nema-v${pkgver//./_}.tar.gz")
sha512sums=('354231d9ac64c484c8ee31a0a8fc7dc3875c757b258ae59ca809b936ba0a8ffafffb69c3f4f0af17d5d36f49c424ec31123da739b5c5a4adeca1a16897cb5c66')

build() {
  cd "$pkgname-v${pkgver//./_}"

  ./configure --prefix=/usr

  make
}

package() {
  cd "$pkgname-v${pkgver//./_}"

  install -Dm 755 $pkgname "$pkgdir/usr/bin/$pkgname"
  install -Dm 644 -t "$pkgdir/usr/share/doc/$pkgname/" README ChangeLog
  install -Dm 644 LICENSE "$pkgdir/usr/share/licenses/$pkgname/LICENSE"
}

